

Anhang zur Arbeit:

***Key-Performance-Indikator
als Qualitätskennzahl
von Handelssystemen***

INHALTSVERZEICHNIS

PROGRAMMCODE	3
INDIKATOR "KPI"	3
KPI-RELEVANTE FUNKTIONEN	8
<i>Funktion "SystemGewinnmitnahmeEffizienz"</i>	8
<i>Funktion "SystemOffenesGewinnVerlustVerhaeltnis"</i>	9
<i>Funktion „MaxDrawupDichte“</i>	10
<i>Funktion "MaxDrawupDrawdownVerhaeltnis"</i>	12
NICHT KPI-RELEVANTE FUNKTIONEN.....	14
<i>Funktion "MaxDrawupRegression"</i>	14
<i>Funktion "MaxDrowdownsRegression"</i>	16
BACKTESTERGEBNISSE FÜR DIE QUALITÄTSUNTERSUCHUNGEN	19
RANKINGS	19
<i>Platz 1</i>	19
<i>Platz 2</i>	20
<i>Platz 3</i>	20
<i>Platz 4</i>	21
<i>Platz 5</i>	21
<i>Platz 6</i>	22
<i>Platz 7</i>	22
<i>Platz 8</i>	23
<i>Platz 9</i>	23
<i>Platz 10</i>	24
<i>Platz 11</i>	24
<i>Platz 12</i>	25
<i>Platz 13</i>	25
<i>Platz 14</i>	26
<i>Platz 15</i>	26
<i>Platz 16</i>	27
<i>Platz 17</i>	27
<i>Platz 18</i>	28
<i>Platz 19</i>	28
<i>Platz 20</i>	29
<i>Platz 21</i>	29
<i>Platz 22</i>	30
<i>Platz 23</i>	30
<i>Platz 24</i>	31
<i>Platz 25</i>	31
<i>Platz 26</i>	32
<i>Platz 27</i>	32
<i>Platz 28</i>	33
<i>Platz 29</i>	33
<i>Platz 30</i>	34
<i>Platz 31</i>	34
<i>Platz 32</i>	35
<i>Platz 33</i>	35

PROGRAMMCODE

Der nun gelistete Equilla-Code kann in Tradesignal importiert werden. Damit kann der KPI praktisch angewendet werden. Der KPI ist als Indikator umgesetzt. Er verwendet für die Berechnung der Aspektwerte entsprechende Funktionsaufrufe.

In den Untersuchungen haben sich leider bestimmte Aspekte als nicht qualitätsaussagend herausgestellt. Zwei Aspekte seien hier trotzdem erwähnt, da diese für weitere Analysen und Forschungen interessant erscheinen. Der Aspekt „MaxDrawupRegression“ befasst sich mit der Idee, die Höhe der jeweils nächsten HE-Position über eine Regressionsanalyse zu schätzen. Weiterhin befasst sich der Aspekt „MaxDrawdownRegression“ mit der Idee den Wert des jeweils nächsten System-MaxDrawdowns zu schätzen (ebenfalls über eine Regressionsanalyse).

INDIKATOR “KPI”

Meta:

```
Categories("Key Performance Indicator; VTAD Award 2017"),
Synopsis("Beitrag zur Arbeit: Key-Performance-Indikator
als Qualitätskennzahl von Handelssystemen"),
Author("Detlev Matthes (dmatthes@web.de)"),
ShortCode("KPI"),
SubChart(True),
WebLink("http://www.vtad.de");
```

```
/*
 * Der KPI verwendet die resultierenden Werte von Aspekten um eine
 * umfassende Kennzahl für das System zu ermitteln.
 *
 * Die Aspektewerte werden durch den Aufruf der
 * folgenden Funktionen ermittelt:
 *     SystemGewinnmitnahmeEffizienz
 *     SystemOffenesGewinnVerlustVerhaeltnis
 *     MaxDrawupDichte
 *     MaxDrawupDrawdownVerhaeltnis
 *     MaxDrawupRegression
 *     MaxDrowdownsRegression
 */
```

Inputs:

```
// Anzeige des KPI und der Aspekt-Werte im Ausgabefenster
printValues(true),

// Zeichne Open Equity
visibleInfoOpenEquity(true),

// Zeichne Closed Equity
visibleInfoClosedEquity(true),

// Zeichne Highest Closed Equity
visibleInfoHighestCloseEquity(true),

// Kennzeichne HE-Positionen und deren Werte
visibleInfoEquityDrawups(false),

// Kennzeichne ClosedMaxDrawdowns und OpenMaxDrawdown und deren Werte
visibleInfoEquityDrawdowns(false),

// Kennzeichne die Werte für abschätzbare nächte HE-Positionen
```

```

visibleNextEstimatedMaxDrawup(false),

// Kennzeichne die Werte für abschätzbare nächste MaxDrawdowns
visibleNextEstimatedMaxDrawdown(false);

Vars:
// Wert für den berechneten KPI
keyPerformance,

// Aspektwert: Gewinnmitnahme-Effizienz (KPI-relevant)
systemGewinnmitnahmeEffizienzValue,

// Aspektwert: offenes GewinnVerlust-Verhältnis (KPI-relevant)
systemOffenesGewinnVerlustVerhaeltnis,

// Aspektwert: MaxDrawupDichte (KPI-relevant)
systemMaxDrawupDichteValue,

// Aspektwert: MaxDrawupBestimmbarkeit (nicht KPI-relevant)
systemDrawupBestimmbarkeit,

// Aspektwert: MaxDrawdownBestimmbarkeit (nicht KPI-relevant)
systemDrawdownBestimmbarkeitValue,

// Aspektwert: MaxDrawupDrawdownVerhaeltnis (KPI-relevant)
systemDrawupDrawdownVerhaeltnisValue,

// Berechnete Werte aus der Regressionsanalyse von HE-Positionen
MaxDrawupRegression_Steigung,
MaxDrawupRegression_Abstand,
MaxDrawupRegression_R,
MaxDrawupRegression_R2,
MaxDrawupRegression_lastX,

// Wert der abschätzbare nächsten HE-Position
nextEstimatedMaxDrawup,

// Berechnete Werte aus der Regressionsanalyse von Maxdrawdowns
MaxDrowdownsRegression_Steigung,
MaxDrowdownsRegression_Abstand,
MaxDrowdownsRegression_R,
MaxDrowdownsRegression_R2,
MaxDrowdownsRegression_lastX,

// Wert des abschätzbaren nächsten MaxDrawdowns
nextEstimatedMaxDrawdown,

// aktuelle höchster Equity-Wert
highestSystemEquity(0),

// aktuelle niedrigster Equity-Wert nach einer HE-Position
minEquity(0),

// letzter ClosedMaxDrawdown
closedMaxDrawdown,

// letzter OpenMaxDrawdown
openMaxDrawdown,

// Marker: Index letzter Position
Data1::lastClosedTradeCount4KPI,

```

```
// Marker: letzter MaxDrawup-Wert bereits gezeigt
nextEstimatedMaxDrawupShown(false),

// Marker: letzter MaxDrawdown-Wert bereits gezeigt
nextEstimatedMaxDrawdownShown(false);
```

Arrays:

```
// closed maximum drawdowns
drawdowns[];
```

```
if Data1::lastClosedTradeCount4KPI <> ClosedTradeCount then begin
    Data1::lastClosedTradeCount4KPI = ClosedTradeCount;
    if ClosedTradeCount > 0 then begin
        if GlobalClosedEquity > highestSystemEquity then begin
            nextEstimatedMaxDrawupShown = true;
            closedMaxDrawdown = highestSystemEquity - minEquity;
            highestSystemEquity = GlobalClosedEquity;
            if visibleInfoEquityDrawups then begin
                DrawSymbol(GlobalClosedEquity, "NewDrawUp",
                    SymbolTriangleUp, 6, Black, Green);
                DrawText(highestSystemEquity * 1.1,
                    "highestSystemEquityValue",
                    "HE:" + highestSystemEquity);
            end;
            if closedMaxDrawdown > 0 then begin
                AppendArray(drawdowns, closedMaxDrawdown);
                if visibleInfoEquityDrawdowns then begin
                    DrawText[-15](minEquity * 0.95,
                        "ClosedMaxDrawdownValue",
                        "CMDD:" + closedMaxDrawdown);
                end;
            end;
            minEquity = highestSystemEquity;
            nextEstimatedMaxDrawdownShown = true;
        end else begin
            if GlobalClosedEquity < minEquity then
                minEquity = GlobalClosedEquity;
            end;
        end else begin
            minEquity = GlobalClosedEquity;
        end;
    end;
end;

// Am letzten Bar den Wert des OpenMaxDrawdown zeigen
if CurrentBar = LastBar then begin
    openMaxDrawdown = highestSystemEquity - minEquity;
    if visibleInfoEquityDrawdowns then begin
        DrawText[25](minEquity * 0.95, "OpenMaxDrawdownValue",
            "OMDD:" + openMaxDrawdown);
    end;
end;

if visibleInfoHighestCloseEquity then
    DrawLine(highestSystemEquity, "Highest Closed Equity",
        StyleDash, 2, Green, Green);

if visibleInfoOpenEquity then
    DrawLine(GlobalOpenEquity, "Open Equity",
        StyleSolid, 2, Blue, Green);
```

```

if visibleInfoClosedEquity then
    DrawLine(GlobalClosedEquity, "Closed Equity",
             StyleSolid, 2, Green, Green);

/*
 * KPI-relevanter Wert: GewinnmitnahmeEffizienz
 */
systemGewinnmitnahmeEffizienzValue = SystemGewinnmitnahmeEffizienz;

/*
 * KPI-relevanter Wert: OffenesGewinnVerlustVerhältnis
 */
systemOffenesGewinnVerlustVerhaeltnis = SystemOffenesGewinnVerlustVerhaeltnis;

/*
 * KPI-relevanter Wert: MaxDrawupDichte
 */
systemMaxDrawupDichteValue = MaxDrawupDichte;

/*
 * KPI-relevanter Wert: MaxDrawupDrawdownVerhältnis
 */
systemDrawupDrawdownVerhaeltnisValue = MaxDrawupDrawdownVerhaeltnis;

MaxDrawupRegression(MaxDrawupRegression_Steigung,
                    MaxDrawupRegression_Abstand,
                    MaxDrawupRegression_R,
                    MaxDrawupRegression_R2,
                    MaxDrawupRegression_lastX);
systemDrawupBestimmbarkeit = MaxDrawupRegression_R2;

/*
 * Bei genügender Bestimmbarkeit (> 95) wird der nächste HE-Position anhand
 * der Regressionsformel geschätzt und angezeigt.
 */
if visibleNextEstimatedMaxDrawup and
    systemDrawupBestimmbarkeit > 0.95 and
    nextEstimatedMaxDrawupShown then begin
    nextEstimatedMaxDrawup = MaxDrawupRegression_Abstand *
        exp((MaxDrawupRegression_lastX+1) *
            MaxDrawupRegression_Steigung);
    if nextEstimatedMaxDrawup > 0 then begin
        DrawSymbol(nextEstimatedMaxDrawup,
                   "nextEstimatedMaxDrawup",
                   SymbolTriangleRight, 12, Green, Red);
        nextEstimatedMaxDrawupShown = false;
    end;
end;

/*
 * nicht KPI-relevanter Wert: MaxDrawdownBestimmbarkeit
 */
MaxDrowdownsRegression(MaxDrowdownsRegression_Steigung,

```

```

        MaxDrawdownsRegression_Abstand,
        MaxDrawdownsRegression_R,
        MaxDrawdownsRegression_R2,
        MaxDrawdownsRegression_lastX);
systemDrawdownBestimmbarkeitValue = MaxDrawdownsRegression_R2;

/*
 * Bei genügender Bestimmbarkeit (> 95) wird der nächste Systemdrawdown anhand
 * der Regressionsformel geschätzt und angezeigt.
 */
if visibleNextEstimatedMaxDrawdown and
    systemDrawdownBestimmbarkeitValue > 0.95 and
    nextEstimatedMaxDrawdownShown then begin
    nextEstimatedMaxDrawdown = MaxDrawdownsRegression_Abstand *
        exp((MaxDrawdownsRegression_lastX+1) *
            MaxDrawdownsRegression_Steigung);
    if nextEstimatedMaxDrawdown > 0 then begin
        DrawSymbol(highestSystemEquity - nextEstimatedMaxDrawdown,
            "nextEstimatedMaxDrawdown",
            SymbolTriangleRight, 12, Black, Red);
        nextEstimatedMaxDrawdownShown = false;
    end;
end;

/*
 * Bestimmung des KPI-Wertes aus den KPI-relevante Aspekten
 */
keyPerformance =
    systemGewinnmitnahmeEffizienzValue +
    systemOffenesGewinnVerlustVerhaeltnis +
    systemMaxDrawupDichteValue +
    systemDrawupDrawdownVerhaeltnisValue;
keyPerformance = keyPerformance / 4.0;

/*
 * Ausgabe der Werte des KPIs und aller KPI-relevanten und
 * nicht KPI-relevanten Aspekte
 */
if printValues then begin
    if CurrentBar = LastBar then begin
        print("Systemwert (KPI):",
            keyPerformance);
        print("Aspekt: GewinnmitnahmeEffizienz",
            systemGewinnmitnahmeEffizienzValue);
        print("Aspekt: OffenesGewinnVerlustVerhaeltnis",
            systemOffenesGewinnVerlustVerhaeltnis);
        print("Aspekt: MaxDrawupDichte",
            systemMaxDrawupDichteValue);
        print("Aspekt: MaxDrawupDrawdownVerhältnis",
            systemDrawupDrawdownVerhaeltnisValue);
        print("Aspekt: MaxDrawupBestimmbarkeit",
            systemDrawupBestimmbarkeit);
        print("Aspekt: MaxDrawdownBestimmbarkeit",
            systemDrawdownBestimmbarkeitValue);
    end;
end;

```

KPI-RELEVANTE FUNKTIONEN

FUNKTION "SYSTEMGEWINNMITNAHMEEFFIZIENZ"

Meta:

```
Categories("Key Performance Indicator; VTAD Award 2017"),
Author("Detlev Matthes (dmatthes@web.de)"),
Synopsis( "Aspekt: SystemGewinnmitnahmeEffizienz" );
```

```
/*
 * Beim Aspekt „Gewinnmitnahme-Effizienz“
 * handelt es sich um das Verhältnis
 * vom ClosedTradeGuV zum MaxOpenTradeGuV.
 *
 * Dieser Aspekt ist KPI-relevant.
 */
```

Vars:

```
// der maximale open equity eines Trades
maximunTradeProfit,

// die Nummer des letzten Trades
Data1::lastClosedTradeCount4SRE(-1),

// Summe aller maximalen open equities von Trades
Data1::maximumSystemProfit4SRE(0);
```

```
if Data1::lastClosedTradeCount4SRE <> ClosedTradeCount then begin
    Data1::lastClosedTradeCount4SRE = ClosedTradeCount;

    maximunTradeProfit = ClosedTradeRunup(ClosedTradeCount)
                        - ClosedTradeEntrySlippage(ClosedTradeCount)
                        - ClosedTradeExitSlippage(ClosedTradeCount)
                        - ClosedTradeEntryCommission(ClosedTradeCount)
                        - ClosedTradeExitCommission(ClosedTradeCount);

    if maximunTradeProfit > 0 then begin
        Data1::maximumSystemProfit4SRE =
            Data1::maximumSystemProfit4SRE + maximunTradeProfit;
    end;

    if Data1::maximumSystemProfit4SRE > 0 then begin
        SystemGewinnmitnahmeEffizienz =
            GlobalClosedEquity / Data1::maximumSystemProfit4SRE;
    end;
end;
```


FUNKTION "SYSTEMOFFENESGEWINNVERLUSTVERHAELTNIS"

Meta:

```
Categories("Key Performance Indicator; VTAD Award 2017"),
Author("Detlev Matthes (dmatthes@web.de)"),
Synopsis("Aspekt: SystemOffenesGewinnVerlustVerhaeltnis");
```

```
/*
 * Beim Aspekt „SystemOffenesGewinnVerlustVerhaeltnis“
 * handelt es sich um den Anteil des OpenTradeGuV innerhalb
 * der Gewinnzone im Verhältnis zur Gesamtzone.
 *
 * Dieser Aspekt ist KPI-relevant.
 */
```

Vars:

```
// der maximale open equity eines Trades
maximumTradeProfit,
```

```
// der minimale open equity eines Trades
maximumTradeLoss,
```

```
// die Nummer des letzten Trades
Data1::lastClosedTradeCount4SGV(-1),
```

```
// Summe aller maximal open Tradeequities
Data1::maximumSystemProfit4SGV(0),
```

```
// Summe aller minimal open Tradeequities
Data1::maximumSystemLoss4SGV(0);
```

```
if Data1::lastClosedTradeCount4SGV <> ClosedTradeCount then begin
    Data1::lastClosedTradeCount4SGV = ClosedTradeCount;

    maximumTradeProfit = ClosedTradeRunup(ClosedTradeCount);
    if maximumTradeProfit > 0 then begin
        Data1::maximumSystemProfit4SGV =
            Data1::maximumSystemProfit4SGV + maximumTradeProfit;
    end;

    maximumTradeLoss = ClosedTradeDrawdown(ClosedTradeCount);
    if maximumTradeLoss < 0 then begin
        Data1::maximumSystemLoss4SGV =
            Data1::maximumSystemLoss4SGV + maximumTradeLoss;
    end;

    if (Data1::maximumSystemProfit4SGV -
        Data1::maximumSystemLoss4SGV) > 0 then begin
        SystemOffenesGewinnVerlustVerhaeltnis =
            Data1::maximumSystemProfit4SGV /
            (Data1::maximumSystemProfit4SGV -
             Data1::maximumSystemLoss4SGV);
    end;
end;
```

FUNKTION „MAXDRAWUPDICHTE“

Meta:

```
Categories("Key Performance Indicator; VTAD Award 2017"),
Author("Detlev Matthes (dmatthes@web.de)"),
Synopsis("Aspekt: MaxDrawupDichte");
```

```
/*
 * Beim Aspekt „MaxDrawupDichte“
 * handelt es sich um die Gleichverteilung von Trades,
 * bei denen eine höhere Equity erreicht wurde.
 *
 * Dieser Aspekt ist KPI-relevant.
 */
```

Vars:

```
// Equity des aktuellen Trades
closedTradeEquity,

// die aktuell höchste Equity des Systems
highestSystemEquity(0),

// die Nummer des letzten Trades
Data1::lastClosedTradeCount4HED(-1),

// Anzahl von Trades bei denen eine höhere Equity erreicht wurde
numberOfX,
mittlererAbstand,
wirklichePosition,
num,
idealPosition,
abstand,
normAbweichungSumme,
abstandListIndex,
absAbweichung,
normAbweichung,

// mittlere normierte Abweichung
abweichung;
```

Arrays:

```
// Array mit Verteilung von Trades bei
// denen eine höhere Equity erreicht wurde
nextHigherEquitys[],

// Array mit Abstände zwischen jeweiligen idealen und wirklichen Index
abstandList[];
```

```
if Data1::lastClosedTradeCount4HED <> ClosedTradeCount then begin
    Data1::lastClosedTradeCount4HED = ClosedTradeCount;
    if ClosedTradeCount > 0 then begin
        if GlobalClosedEquity > highestSystemEquity then begin
            highestSystemEquity = GlobalClosedEquity;
            AppendArray(nextHigherEquitys, true);
            numberOfX = numberOfX + 1;
        end else begin
            AppendArray(nextHigherEquitys, false);
        end;
    end;
end;
```

```

if numberOfX > 1 then begin
    mittlererAbstand = ArrayLength(nextHigherEquitys) / numberOfX;

    num = 0;

    // IHE-Positionen und deren Abstand zur entsprechenden
    // HE-Position berechnen
    ClearArray(abstandList);
    for wirklichePosition = 1 to ArrayLength(nextHigherEquitys) begin
        if nextHigherEquitys[wirklichePosition] = true then begin
            num = num + 1;
            idealPosition = num * mittlererAbstand;
            abstand = wirklichePosition - idealPosition;
            AppendArray(abstandList, abstand);
        end;
    end;

    if ArrayLength(abstandList) > 0 then begin
        // mittlerer normierte Abweichung berechnen
        normAbweichungSumme = 0;
        for abstandListIndex = 1 to ArrayLength(abstandList) begin
            abstand = abstandList[abstandListIndex];
            absAbweichung = abs(abstand);
            normAbweichung = absAbweichung;
            normAbweichungSumme =
                normAbweichungSumme + normAbweichung;
        end;
        abweichung = normAbweichungSumme /
            ArrayLength(nextHigherEquitys);
        abweichung = abweichung /
            ArrayLength(abstandList);

        MaxDrawupDichte = 1 - abweichung;
    end else
        MaxDrawupDichte = 0;
    end;
end;
end;
end;

```

FUNKTION "MAXDRAWUPDRAWDOWNVERHAELTNIS"

Meta:

```
Categories("Key Performance Indicator; VTAD Award 2017"),
Author("Detlev Matthes (dmatthes@web.de)"),
Synopsis("Aspekt: MaxDrawupDrawdownVerhaeltnis" );
```

```
/*
 * Beim Aspekt „MaxDrawupDrawdownVerhaeltnis“
 * befasst sich mit dem kleinsten Verhältnis der
 * Drawdowns zur letzten Gewinnsteigung.
 * Dieser Aspekt ist KPI-relevant.
 */
```

Vars:

```
// die Nummer des letzten Trades
Data1::lastClosedTradeCount4DDDR(-1),

// aktuelle kleinste Equity
minEquity,

// Abstand aktuell höchste Equity des Systems zu minEquity
newDrawdown,

// das aktuell letzte Drawdown
lastDrawdown,

// resultierendes Systemverhältnis
ratio,

// Gesamtsumme aller Drawdowns
summeDrawdowns,

// die aktuell höchste Equity des Systems
highestSystemEquity(0);
```

Arrays:

```
// Array mit allen ClosedMaxDrawdowns
drawdowns[];
```

```
if Data1::lastClosedTradeCount4DDDR <> ClosedTradeCount then begin
    Data1::lastClosedTradeCount4DDDR = ClosedTradeCount;
    if ClosedTradeCount > 0 then begin
        if GlobalClosedEquity > highestSystemEquity then begin
            newDrawdown = highestSystemEquity - minEquity;
            highestSystemEquity = GlobalClosedEquity;
            if newDrawdown > 0 then begin
                AppendArray(drawdowns, newDrawdown);
            end;

            minEquity = highestSystemEquity;
        end else begin
            if GlobalClosedEquity < minEquity then
                minEquity = GlobalClosedEquity;
        end;
    end else begin
        minEquity = GlobalClosedEquity;
    end;
```

```
end;  
  
lastDrawdown = highestSystemEquity - minEquity;  
  
if highestSystemEquity > 0 then begin  
    summeDrawdowns =  
        Array_Sum(drawdowns, 1, ArrayLength(drawdowns)) + lastDrawdown;  
    ratio = highestSystemEquity / (highestSystemEquity + summeDrawdowns);  
    MaxDrawupDrawdownVerhaeltnis = ratio;  
end;  
  
end;
```

NICHT KPI-RELEVANTE FUNKTIONEN

FUNKTION "MAXDRAWUPREGRESSION"

Meta:

```
Categories("Key Performance Indicator; VTAD Award 2017"),
Author("Detlev Matthes (dmatthes@web.de)",
Synopsis( "Aspekt: MaxDrawupRegression" );
```

```
/*
 * Beim Aspekt „MaxDrawupRegression“ handelt es sich um
 * Regression der Trades, bei denen eine höhere Equity erreicht wurde.
 * Für die Regression wird eine Exponentialfunktion der HE-Positionen
 * über deren fortlaufenden Indexes ihres Auftretens angenommen.
 *
 * Zur Durchführung einer Regression mit einer Exponentialfunktion wird eine
 * lineare Regression mit entsprechender Transformation vorgenommen:
 *  $y = a \cdot e^{(bx)} \rightarrow \ln y = \ln a + bx$ 
 *
 * Anwendungsmöglichkeit: Bei entsprechend ausreichender Bestimmbarkeit
 * kann die Höhe der nächsten HE-Position abgeschätzt werden.
 *
 * Dieser Aspekt ist nicht KPI-relevant.
 */
```

Inputs:

```
// Returnvalue: Steigung bei lineare Regression
RegressionResult_Steigung(NumericRef),

// Returnvalue: Achsenabschnitt bei lineare Regression
RegressionResult_Abstand(NumericRef),

// Returnvalue: ermittelter Korrelationskoeffizient der lineare Regression
RegressionResult_R(NumericRef),

// Returnvalue: ermittelte Bestimmtheit der lineare Regression
RegressionResult_R2(NumericRef),

// Returnvalue: Anzahl von HE-Positionen, die bei der Regression
// verwendet wurde
RegressionResult_lastX(NumericRef);
```

Vars:

```
// die aktuell höchste Equity des Systems
highestSystemEquity(0),

highestSystemEquityIndex,

// Anzahl Werte des Arrays nextHigherEquitys
nextHigherEquitysLength,

// Durchschnitt der X-Werte (Index der HE-Position)
dx,

// Durchschnitt der Y-Werte (Wert der HE-Position)
dy,

sum__xi_dx__yi_dy,
sum__xi_xy_2,
sum__yi_dy_2,
```

```
xi_dx,
yi_dy;
```

Arrays:

```
// Array mit Verteilung von Trades bei
// denen eine höhere Equity erreicht wurde
nextHigherEquitys[];
```

```
if Data1::lastClosedTradeCount4HER <> ClosedTradeCount then begin
    Data1::lastClosedTradeCount4HER = ClosedTradeCount;
    if ClosedTradeCount > 0 then begin
        if GlobalClosedEquity > highestSystemEquity then begin
            highestSystemEquity = GlobalClosedEquity;
            AppendArray(nextHigherEquitys, highestSystemEquity);
        end;

        nextHigherEquitysLength = ArrayLength(nextHigherEquitys);

        if nextHigherEquitysLength > 2 then begin
            dx = 0;
            dy = 0;
            for highestSystemEquityIndex = 1 to nextHigherEquitysLength begin
                dx = dx + highestSystemEquityIndex;
                dy = dy + log(nextHigherEquitys[highestSystemEquityIndex]);
            end;

            dx = dx / nextHigherEquitysLength;
            dy = dy / nextHigherEquitysLength;
            sum__xi_dx__yi_dy = 0;
            sum__xi_xy_2 = 0;
            sum__yi_dy_2 = 0;
            for highestSystemEquityIndex = 1 to nextHigherEquitysLength begin
                xi_dx = highestSystemEquityIndex - dx;
                yi_dy = log(nextHigherEquitys[highestSystemEquityIndex]) - dy;
                sum__xi_dx__yi_dy = sum__xi_dx__yi_dy + xi_dx * yi_dy;
                sum__xi_xy_2 = sum__xi_xy_2 + xi_dx * xi_dx;
                sum__yi_dy_2 = sum__yi_dy_2 + yi_dy * yi_dy;
            end;

            RegressionResult_Steigung = sum__xi_dx__yi_dy / sum__xi_xy_2;
            RegressionResult_Abstand = exp(dy - RegressionResult_Steigung * dx);
            RegressionResult_R = sum__xi_dx__yi_dy /
                sqrt(sum__xi_xy_2 * sum__yi_dy_2);
            RegressionResult_R2 = RegressionResult_R * RegressionResult_R;

            MaxDrawupRegression = RegressionResult_R2;
            RegressionResult_lastX = nextHigherEquitysLength;
        end;
    end;
end;
```

FUNKTION "MAXDROWDOWNSREGRESSION"

Meta:

```
Categories("Key Performance Indicator; VTAD Award 2017"),
Author("Detlev Matthes (dmatthes@web.de)"),
Synopsis("Aspekt: MaxDrawdownRegression");
```

```
/*
 * Beim Aspekt „MaxDrawdownRegression“ handelt es sich um
 * Regression der ClosedMaxDrawdowns, die jeweils das höchste
 * System-MaxDrowdown verursacht haben.
 * Für die Regression wird eine Exponentialfunktion der Drawdownwerte
 * über deren fortlaufenden Indexes ihres Auftretens angenommen.
 *
 * Zur Durchführung einer Regression mit einer Exponentialfunktion wird eine
 * lineare Regression mit entsprechender Transformation vorgenommen:
 *  $y = a \cdot e^{(bx)} \rightarrow \ln y = \ln a + bx$ 
 *
 * Anwendungsmöglichkeit: Bei entsprechend ausreichender Bestimmbarkeit
 * kann die Höhe der nächsten System-MaxDrawdowns (letzter OpenMaxDrawdown)
 * abgeschätzt werden.
 *
 * Dieser Aspekt ist nicht KPI-relevant.
 */
```

Inputs:

```
// Returnvalue: Steigung bei lineare Regression
RegressionResult_Steigung(NumericRef),

// Returnvalue: Achsenabschnitt bei lineare Regression
RegressionResult_Abstand(NumericRef),

// Returnvalue: ermittelter Korrelationskoeffizient der lineare Regression
RegressionResult_R(NumericRef),

// Returnvalue: ermittelte Bestimmtheit der lineare Regression
RegressionResult_R2(NumericRef),

// Returnvalue: Anzahl von ClosedMaxDrawdowns, die bei der Regression
// verwendet wurde
RegressionResult_lastX(NumericRef);
```

Vars:

```
Data1::lastClosedTradeCount4HEDR(0),
minEquity,
newDrawdown,
lastMaxDrawdown(0),
drawDownsLength,
drawDownsIndex,
highestSystemEquity(0),

// Durchschnitt der X-Werte (Index der ClosedMaxDrawdowns)
dx,

// Durchschnitt der Y-Werte (Wert der ClosedMaxDrawdowns)
dy,

sum__xi_dx__yi_dy,
sum__xi_xy_2,
sum__yi_dy_2,
```



```
xi_dx,
yi_dy;
```

Arrays:

```
// Array mit ClosedMaxDrawdowns, die das
// jeweils höchste System-Drawdown verursacht haben
drawDowns[];
```

```
if Data1::lastClosedTradeCount4HEDR <> ClosedTradeCount then begin
    Data1::lastClosedTradeCount4HEDR = ClosedTradeCount;
    if ClosedTradeCount > 0 then begin
        if GlobalClosedEquity > highestSystemEquity then begin
            newDrawdown = highestSystemEquity - minEquity;
            if newDrawdown > lastMaxDrawdown then begin
                lastMaxDrawdown = newDrawdown;

                AppendArray(drawDowns, newDrawdown);

                drawDownsLength = ArrayLength(drawDowns);

                if drawDownsLength > 2 then begin
                    dx = 0;
                    dy = 0;
                    for drawDownsIndex = 1 to drawDownsLength begin
                        dx = dx + drawDownsIndex;
                        dy = dy + log(drawDowns[drawDownsIndex]);
                    end;

                    dx = dx / drawDownsLength;
                    dy = dy / drawDownsLength;
                    sum__xi_dx__yi_dy = 0;
                    sum__xi_xy_2 = 0;
                    sum__yi_dy_2 = 0;
                    for drawDownsIndex = 1 to drawDownsLength begin
                        xi_dx = drawDownsIndex - dx;
                        yi_dy = log(drawDowns[drawDownsIndex]) - dy;
                        sum__xi_dx__yi_dy =
                            sum__xi_dx__yi_dy + xi_dx * yi_dy;
                        sum__xi_xy_2 = sum__xi_xy_2 + xi_dx * xi_dx;
                        sum__yi_dy_2 = sum__yi_dy_2 + yi_dy * yi_dy;
                    end;
                    RegressionResult_Steigung =
                        sum__xi_dx__yi_dy / sum__xi_xy_2;
                    RegressionResult_Abstand =
                        exp(dy - RegressionResult_Steigung * dx);
                    RegressionResult_R =
                        sum__xi_dx__yi_dy /
                        sqrt(sum__xi_xy_2 * sum__yi_dy_2);
                    RegressionResult_R2 =
                        RegressionResult_R * RegressionResult_R;

                    MaxDrowdownsRegression = RegressionResult_R2;
                    RegressionResult_lastX = drawDownsLength;
                end;
            end;
            highestSystemEquity = GlobalClosedEquity;
            minEquity = highestSystemEquity;
        end
        else if GlobalClosedEquity < minEquity then
            minEquity = GlobalClosedEquity;
        end;
    end;
end;
```

end;

BACKTESTERGEBNISSE FÜR DIE QUALITÄTSUNTERSUCHUNGEN

Im Folgenden werden die Backtestergebnisse gezeigt, die für die Qualitätsuntersuchungen verwendet wurden.

Die Backtestergebnisse liefern für unterschiedliche Aktien des DAX unterschiedliche Werte und Equity-Verläufe. Bei der Auswahl der Backtestergebnisse wurde auf eine gute Mischung von qualitativ hohen bis -niedrigen Ergebnissen geachtet. Daher war es auch notwendig die Backtests auf verschiedene Zeitebenen durchzuführen.

Die Ergebnisse basieren auf der gleichen Kapitaleinlage, dem gleichen Moneymanagement, dem historischen Zeitraum vom 01.01.1997 bis 21.12.2016. Es gelten die folgenden Handelsregeln:

- Einstieg: MACD kreuzt MACD-Trigger nach Oben (LONG)
- Ausstieg: MACD kreuzt MACD-Trigger nach Unten, Close durchbricht SAR

Diese Backtestergebnisse zeigen den jeweiligen Equityverlauf, Werte für KPI-relevante Aspekte und ausgewählte Kennzahlen.


Bei den zusätzlich genannten Werten „MaxDrawupBestimmbarkeit“ und „MaxDrawdownBestimmbarkeit“ handelt es sich um Ergebnisse von Regressionsanalysen der nicht-KPI-relevanten Aspekte „MaxDrawupRegression“ und „MaxDrawdownRegression“ (siehe „nicht KPI-relevante Funktionen“).

RANKINGS

Die folgende Ranking-Übersicht zeigt das Ergebnis der visuellen Begutachtung und nach Qualität eingeschätzten Backtestergebnissen. Auf den ersten Platz befindet sich das Ergebnis mit der besten Begutachtung und Einschätzung.

PLATZ 1


SAP SE O.N. (wöchentlich)



Systemwert (KPI):	0,717	TotalNetProfit :	8.708,00
Aspekt GewinnmitnahmeEffizienz:	0,354	Profitfaktor:	4,43
Aspekt OffenesGewinnVerlustVerhältnis:	0,768	Trefferquote:	63,41%
Aspekt MaxDrawupDichte:	0,945	Fröhlich-Faktor:	10,80
Aspekt MaxDrawupDrawdownVerhältnis:	0,801		
Aspekt MaxDrawupBestimmbarkeit:	0,534		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 2


THYSSENKRUPP AG O.N. (wöchentlich)



Systemwert (KPI):	0,659	TotalNetProfit:	3.767,00
Aspekt GewinnmitnahmeEffizienz:	0,255	Profitfaktor:	2,46
Aspekt OffenesGewinnVerlustVerhältnis:	0,737	Trefferquote:	60,98%
Aspekt MaxDrawupDichte:	0,954	Fröhlich-Faktor:	4,98
Aspekt MaxDrawupDrawdownVerhältnis:	0,688		
Aspekt MaxDrawupBestimmbarkeit:	0,631		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 3


VOLKSWAGEN AG VZO O.N. (täglich)



Systemwert (KPI):	0,536	TotalNetProfit:	13.731,00
Aspekt GewinnmitnahmeEffizienz:	0,136	Profitfaktor:	1,41
Aspekt OffenesGewinnVerlustVerhältnis:	0,639	Trefferquote:	45,74%
Aspekt MaxDrawupDichte:	0,940	Fröhlich-Faktor:	1,31
Aspekt MaxDrawupDrawdownVerhältnis:	0,431		
Aspekt MaxDrawupBestimmbarkeit:	0,946		
Aspekt MaxDrawdownBestimmbarkeit:	0,826		

PLATZ 4


FRESENIUS SE+CO.KGAA O.N. (täglich)



Systemwert (KPI):	0,555	TotalNetProfit:	2.875,00
Aspekt GewinnmitnahmeEffizienz:	0,159	Profitfaktor:	1,54
Aspekt OffenesGewinnVerlustVerhältnis:	0,647	Trefferquote:	48,11%
Aspekt MaxDrawupDichte:	0,950	Fröhlich-Faktor:	2,35
Aspekt MaxDrawupDrawdownVerhältnis:	0,464		
Aspekt MaxDrawupBestimmbarkeit:	0,702		
Aspekt MaxDrawdownBestimmbarkeit:	0,963		

PLATZ 5


SAP SE O.N. (täglich)



Systemwert (KPI):	0,451	TotalNetProfit:	2.401,00
Aspekt GewinnmitnahmeEffizienz:	0,054	Profitfaktor:	1,13
Aspekt OffenesGewinnVerlustVerhältnis:	0,597	Trefferquote:	41,12%
Aspekt MaxDrawupDichte:	0,874	Fröhlich-Faktor:	0,61
Aspekt MaxDrawupDrawdownVerhältnis:	0,278		
Aspekt MaxDrawupBestimmbarkeit:	0,882		
Aspekt MaxDrawdownBestimmbarkeit:	0,842		

PLATZ 6


BASF SE NA O.N. (täglich)



Systemwert (KPI):	0,450	TotalNetProfit:	884,00
Aspekt GewinnmitnahmeEffizienz:	0,024	Profitfaktor:	1,06
Aspekt OffenesGewinnVerlustVerhältnis:	0,595	Trefferquote:	42,27%
Aspekt MaxDrawupDichte:	0,935	Fröhlich-Faktor:	0,21
Aspekt MaxDrawupDrawdownVerhältnis:	0,247		
Aspekt MaxDrawupBestimmbarkeit:	0,865		
Aspekt MaxDrawdownBestimmbarkeit:	0,956		

PLATZ 7


ADIDAS AG NA O.N. (täglich)



Systemwert (KPI):	0,438	TotalNetProfit:	2.910,00
Aspekt GewinnmitnahmeEffizienz:	0,073	Profitfaktor:	1,21
Aspekt OffenesGewinnVerlustVerhältnis:	0,594	Trefferquote:	38,94%
Aspekt MaxDrawupDichte:	0,791	Fröhlich-Faktor:	1,12
Aspekt MaxDrawupDrawdownVerhältnis:	0,293		
Aspekt MaxDrawupBestimmbarkeit:	0,901		
Aspekt MaxDrawdownBestimmbarkeit:	0,631		

PLATZ 8


VOLKSWAGEN AG VZO O.N. (wöchentlich)



Systemwert (KPI):	0,525	TotalNetProfit:	6.965,00
Aspekt GewinnmitnahmeEffizienz:	0,144	Profitfaktor:	1,44
Aspekt OffenesGewinnVerlustVerhältnis:	0,663	Trefferquote:	38,10%
Aspekt MaxDrawupDichte:	0,881	Fröhlich-Faktor:	0,72
Aspekt MaxDrawupDrawdownVerhältnis:	0,414		
Aspekt MaxDrawupBestimmbarkeit:	0,945		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 9


THYSSENKRUPP AG O.N. (täglich)



Systemwert (KPI):	0,381	TotalNetProfit:	269,50
Aspekt GewinnmitnahmeEffizienz:	0,010	Profitfaktor:	1,03
Aspekt OffenesGewinnVerlustVerhältnis:	0,616	Trefferquote:	42,25%
Aspekt MaxDrawupDichte:	0,733	Fröhlich-Faktor:	0,11
Aspekt MaxDrawupDrawdownVerhältnis:	0,165		
Aspekt MaxDrawupBestimmbarkeit:	0,903		
Aspekt MaxDrawdownBestimmbarkeit:	0,830		

PLATZ 10


DEUTSCHE POST AG NA O.N. (täglich)



Systemwert (KPI):	0,411	TotalNetProfit:	565,60
Aspekt GewinnmitnahmeEffizienz:	0,043	Profitfaktor:	1,11
Aspekt OffenesGewinnVerlustVerhältnis:	0,623	Trefferquote:	40,24%
Aspekt MaxDrawupDichte:	0,685	Fröhlich-Faktor:	0,46
Aspekt MaxDrawupDrawdownVerhältnis:	0,293		
Aspekt MaxDrawupBestimmbarkeit:	0,968		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 11


COMMERZBANK AG (täglich)



Systemwert (KPI):	0,475	TotalNetProfit:	15.923,80
Aspekt GewinnmitnahmeEffizienz:	0,084	Profitfaktor:	1,23
Aspekt OffenesGewinnVerlustVerhältnis:	0,637	Trefferquote:	40,93%
Aspekt MaxDrawupDichte:	0,769	Fröhlich-Faktor:	0,86
Aspekt MaxDrawupDrawdownVerhältnis:	0,412		
Aspekt MaxDrawupBestimmbarkeit:	0,813		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 12


BAYER AG NA O.N. (täglich)



Systemwert (KPI):	0,433	TotalNetProfit:	437,00
Aspekt GewinnmitnahmeEffizienz:	0,008	Profitfaktor:	1,02
Aspekt OffenesGewinnVerlustVerhältnis:	0,587	Trefferquote:	37,05%
Aspekt MaxDrawupDichte:	0,846	Fröhlich-Faktor:	0,07
Aspekt MaxDrawupDrawdownVerhältnis:	0,292		
Aspekt MaxDrawupBestimmbarkeit:	0,953		
Aspekt MaxDrawdownBestimmbarkeit:	0,846		

PLATZ 13


CONTINENTAL AG O.N. (täglich)



Systemwert (KPI):	0,414	TotalNetProfit:	-4.350,00
Aspekt GewinnmitnahmeEffizienz:	-0,059	Profitfaktor:	0,86
Aspekt OffenesGewinnVerlustVerhältnis:	0,589	Trefferquote:	39,39%
Aspekt MaxDrawupDichte:	0,887	Fröhlich-Faktor:	-0,26
Aspekt MaxDrawupDrawdownVerhältnis:	0,238		
Aspekt MaxDrawupBestimmbarkeit:	0,910		
Aspekt MaxDrawdownBestimmbarkeit:	0,995		

PLATZ 14


DAIMLER AG NA O.N. (täglich)



Systemwert (KPI):	0,425	TotalNetProfit:	-352,00
Aspekt GewinnmitnahmeEffizienz:	-0,006	Profitfaktor:	0,98
Aspekt OffenesGewinnVerlustVerhältnis:	0,613	Trefferquote:	37,38%
Aspekt MaxDrawupDichte:	0,865	Fröhlich-Faktor:	-0,05
Aspekt MaxDrawupDrawdownVerhältnis:	0,229		
Aspekt MaxDrawupBestimmbarkeit:	0,970		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 15


RWE AG ST O.N. (täglich)



Systemwert (KPI):	0,440	TotalNetProfit:	1.251,30
Aspekt GewinnmitnahmeEffizienz:	0,028	Profitfaktor:	1,07
Aspekt OffenesGewinnVerlustVerhältnis:	0,617	Trefferquote:	42,61%
Aspekt MaxDrawupDichte:	0,785	Fröhlich-Faktor:	0,36
Aspekt MaxDrawupDrawdownVerhältnis:	0,330		
Aspekt MaxDrawupBestimmbarkeit:	0,894		
Aspekt MaxDrawdownBestimmbarkeit:	0,896		

PLATZ 16

MERCK KGAA O.N. (täglich)



Systemwert (KPI):	0,455	TotalNetProfit:	2.531,00
Aspekt GewinnmitnahmeEffizienz:	0,066	Profitfaktor:	1,17
Aspekt OffenesGewinnVerlustVerhältnis:	0,606	Trefferquote:	35,91%
Aspekt MaxDrawupDichte:	0,873	Fröhlich-Faktor:	0,55
Aspekt MaxDrawupDrawdownVerhältnis:	0,274		
Aspekt MaxDrawupBestimmbarkeit:	0,967		
Aspekt MaxDrawdownBestimmbarkeit:	0,865		

PLATZ 17


SIEMENS AG NA (täglich)



Systemwert (KPI):	0,426	TotalNetProfit:	4.201,00
Aspekt GewinnmitnahmeEffizienz:	0,054	Profitfaktor:	1,16
Aspekt OffenesGewinnVerlustVerhältnis:	0,642	Trefferquote:	43,06%
Aspekt MaxDrawupDichte:	0,585	Fröhlich-Faktor:	0,61
Aspekt MaxDrawupDrawdownVerhältnis:	0,425		
Aspekt MaxDrawupBestimmbarkeit:	0,825		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 18


HENKEL AG+CO KGAA VZO (täglich)



Systemwert (KPI):	0,408	TotalNetProfit:	510,00
Aspekt GewinnmitnahmeEffizienz:	0,014	Profitfaktor:	1,04
Aspekt OffenesGewinnVerlustVerhältnis:	0,604	Trefferquote:	41,51%
Aspekt MaxDrawupDichte:	0,676	Fröhlich-Faktor:	0,11
Aspekt MaxDrawupDrawdownVerhältnis:	0,340		
Aspekt MaxDrawupBestimmbarkeit:	0,971		
Aspekt MaxDrawdownBestimmbarkeit:	1,000		

PLATZ 19


LINDE AG O.N. (täglich)



Systemwert (KPI):	0,381	TotalNetProfit:	-796,00
Aspekt GewinnmitnahmeEffizienz:	-0,010	Profitfaktor:	0,97
Aspekt OffenesGewinnVerlustVerhältnis:	0,594	Trefferquote:	45,06%
Aspekt MaxDrawupDichte:	0,780	Fröhlich-Faktor:	-0,08
Aspekt MaxDrawupDrawdownVerhältnis:	0,160		
Aspekt MaxDrawupBestimmbarkeit:	0,826		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 20


ALLIANZ SE NA O.N. (täglich)



Systemwert (KPI):	0,401	TotalNetProfit:	10.082,00
Aspekt GewinnmitnahmeEffizienz:	0,056	Profitfaktor:	1,15
Aspekt OffenesGewinnVerlustVerhältnis:	0,613	Trefferquote:	43,50%
Aspekt MaxDrawupDichte:	0,527	Fröhlich-Faktor:	0,33
Aspekt MaxDrawupDrawdownVerhältnis:	0,408		
Aspekt MaxDrawupBestimmbarkeit:	0,974		
Aspekt MaxDrawdownBestimmbarkeit:	0,804		

PLATZ 21


MUENCH.RUECKVERS.VNA O.N. (täglich)



Systemwert (KPI):	0,413	TotalNetProfit:	7.612,00
Aspekt GewinnmitnahmeEffizienz:	0,043	Profitfaktor:	1,11
Aspekt OffenesGewinnVerlustVerhältnis:	0,615	Trefferquote:	41,06%
Aspekt MaxDrawupDichte:	0,510	Fröhlich-Faktor:	0,29
Aspekt MaxDrawupDrawdownVerhältnis:	0,483		
Aspekt MaxDrawupBestimmbarkeit:	0,802		
Aspekt MaxDrawdownBestimmbarkeit:	0,952		

PLATZ 22


BAY.MOTOREN WERKE AG ST (täglich)



Systemwert (KPI):	0,350	TotalNetProfit:	-1.244,00
Aspekt GewinnmitnahmeEffizienz:	-0,024	Profitfaktor:	0,94
Aspekt OffenesGewinnVerlustVerhältnis:	0,600	Trefferquote:	37,85%
Aspekt MaxDrawupDichte:	0,625	Fröhlich-Faktor:	-0,19
Aspekt MaxDrawupDrawdownVerhältnis:	0,197		
Aspekt MaxDrawupBestimmbarkeit:	0,959		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 23


BEIERSDORF AG O.N. (täglich)



Systemwert (KPI):	0,259	TotalNetProfit:	-2.236,00
Aspekt GewinnmitnahmeEffizienz:	-0,059	Profitfaktor:	0,86
Aspekt OffenesGewinnVerlustVerhältnis:	0,577	Trefferquote:	34,11%
Aspekt MaxDrawupDichte:	0,431	Fröhlich-Faktor:	-0,43
Aspekt MaxDrawupDrawdownVerhältnis:	0,087		
Aspekt MaxDrawupBestimmbarkeit:	0,848		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 24


SIEMENS AG NA (wöchentlich)



Systemwert (KPI):	0,368	TotalNetProfit:	2.402,00
Aspekt GewinnmitnahmeEffizienz:	0,068	Profitfaktor:	1,17
Aspekt OffenesGewinnVerlustVerhältnis:	0,603	Trefferquote:	40,00%
Aspekt MaxDrawupDichte:	0,350	Fröhlich-Faktor:	0,23
Aspekt MaxDrawupDrawdownVerhältnis:	0,450		
Aspekt MaxDrawupBestimmbarkeit:	0,000		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 25


VONOVIA SE NA O.N. (täglich)



Systemwert (KPI):	0,357	TotalNetProfit:	-216,00
Aspekt GewinnmitnahmeEffizienz:	-0,060	Profitfaktor:	0,88
Aspekt OffenesGewinnVerlustVerhältnis:	0,533	Trefferquote:	32,56%
Aspekt MaxDrawupDichte:	0,749	Fröhlich-Faktor:	-0,13
Aspekt MaxDrawupDrawdownVerhältnis:	0,206		
Aspekt MaxDrawupBestimmbarkeit:	0,935		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 26


FRESEN.MED.CARE KGAA O.N. (täglich)



Systemwert (KPI):	0,382	TotalNetProfit:	-1.170,00
Aspekt GewinnmitnahmeEffizienz:	-0,038	Profitfaktor:	0,91
Aspekt OffenesGewinnVerlustVerhältnis:	0,582	Trefferquote:	35,81%
Aspekt MaxDrawupDichte:	0,833	Fröhlich-Faktor:	-0,32
Aspekt MaxDrawupDrawdownVerhältnis:	0,150		
Aspekt MaxDrawupBestimmbarkeit:	0,835		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 27

HEIDELBERGCEMENT ST O.N. (täglich)



Systemwert (KPI):	0,312	TotalNetProfit:	-733,00
Aspekt GewinnmitnahmeEffizienz:	-0,011	Profitfaktor:	0,97
Aspekt OffenesGewinnVerlustVerhältnis:	0,592	Trefferquote:	38,35%
Aspekt MaxDrawupDichte:	0,512	Fröhlich-Faktor:	-0,08
Aspekt MaxDrawupDrawdownVerhältnis:	0,157		
Aspekt MaxDrawupBestimmbarkeit:	0,927		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 28

INFINEON TECH.AG NA O.N. (täglich)



Systemwert (KPI):	0,224	TotalNetProfit:	-2.994,20
Aspekt GewinnmitnahmeEffizienz:	-0,173	Profitfaktor:	0,64
Aspekt OffenesGewinnVerlustVerhältnis:	0,557	Trefferquote:	36,78%
Aspekt MaxDrawupDichte:	0,259	Fröhlich-Faktor:	-0,30
Aspekt MaxDrawupDrawdownVerhältnis:	0,253		
Aspekt MaxDrawupBestimmbarkeit:	0,000		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 29

LUFTHANSA AG VNA O.N. (täglich)



Systemwert (KPI):	0,246	TotalNetProfit:	-1.396,30
Aspekt GewinnmitnahmeEffizienz:	-0,077	Profitfaktor:	0,83
Aspekt OffenesGewinnVerlustVerhältnis:	0,575	Trefferquote:	36,40%
Aspekt MaxDrawupDichte:	0,351	Fröhlich-Faktor:	-0,48
Aspekt MaxDrawupDrawdownVerhältnis:	0,136		
Aspekt MaxDrawupBestimmbarkeit:	0,926		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 30


E.ON SE NA O.N. (täglich)



Systemwert (KPI):	0,234	TotalNetProfit:	-1.889,20
Aspekt GewinnmitnahmeEffizienz:	-0,092	Profitfaktor:	0,79
Aspekt OffenesGewinnVerlustVerhältnis:	0,580	Trefferquote:	38,84%
Aspekt MaxDrawupDichte:	0,395	Fröhlich-Faktor:	-0,48
Aspekt MaxDrawupDrawdownVerhältnis:	0,054		
Aspekt MaxDrawupBestimmbarkeit:	0,782		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 31


DEUTSCHE TELEKOM AG NA (täglich)



Systemwert (KPI):	0,277	TotalNetProfit:	-2.115,40
Aspekt GewinnmitnahmeEffizienz:	-0,097	Profitfaktor:	0,79
Aspekt OffenesGewinnVerlustVerhältnis:	0,565	Trefferquote:	35,71%
Aspekt MaxDrawupDichte:	0,500	Fröhlich-Faktor:	-0,30
Aspekt MaxDrawupDrawdownVerhältnis:	0,142		
Aspekt MaxDrawupBestimmbarkeit:	0,649		
Aspekt MaxDrawdownBestimmbarkeit:	0,904		

PLATZ 32


DEUTSCHE BANK AG NA O.N. (täglich)



Systemwert (KPI):	0,326	TotalNetProfit:	-3.990,50
Aspekt GewinnmitnahmeEffizienz:	-0,065	Profitfaktor:	0,85
Aspekt OffenesGewinnVerlustVerhältnis:	0,586	Trefferquote:	38,00%
Aspekt MaxDrawupDichte:	0,512	Fröhlich-Faktor:	-0,31
Aspekt MaxDrawupDrawdownVerhältnis:	0,271		
Aspekt MaxDrawupBestimmbarkeit:	0,938		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		

PLATZ 33

PROSIEBENSAT.1 NA O.N. (täglich)



Systemwert (KPI):	0,181	TotalNetProfit:	-4.769,50
Aspekt GewinnmitnahmeEffizienz:	-0,181	Profitfaktor:	0,64
Aspekt OffenesGewinnVerlustVerhältnis:	0,553	Trefferquote:	33,33%
Aspekt MaxDrawupDichte:	0,319	Fröhlich-Faktor:	-0,52
Aspekt MaxDrawupDrawdownVerhältnis:	0,034		
Aspekt MaxDrawupBestimmbarkeit:	0,000		
Aspekt MaxDrawdownBestimmbarkeit:	0,000		